
Grails 2.0 - 1.4にしておくにはもったいない！
今から半年ほど前の 2011 年 6 月 27 日。Grails 開発チームの発

表にあったように、1.4 系にしておくのはもったいないとのこと
で、Grails1.4 系は 2.0 系となりました。詳細は JGGUG の G* ワー
クショップにて解説された tyama さんのスライドをご覧くださ
い。

さてどの辺りが「もったいない！」程の内容なのかを探るシリー
ズ第二弾ですが、企画開始 ( 前々号 ) から、今回分を ( 多忙な日々
をかいくぐり ) 執筆してる間に、ついに 2011 年 12 月 15 日 (JST) に、
Grails-2.0 が正式リリースされました！

このシリーズは、今後 「Grails 2.0 の世界」として継続していく
" ハズ " です！

では、今回は、大きな変更点の一つ「Resources プラグイン」
を通称 " ヒゲのおっさん " がお送りします！

■Resources プラグイン 
Grails-2.0 には、今回の目玉商品である "Resources プラグイン "

がデフォルトでインストールされています。Grails 2.0.0には、バー
ジョン 1.1.6 の Resources プラグインがインストールされます。こ
の Resouces プラグインとは静的リソースを管理する為の " リソー
スフレームワークプラグイン " です。

通常の Grails アプリケーションではリソース指定のタグを利用
してリソースを管理 ( 指定 ) していました。しかし、最近の Web
アプリケーションは JavaScript ＆ CSS のライブラリやフレーム
ワークに依存して開発することが多くなってきているため、管理
などが複雑で大変になってきます。

そこで、Resources プラグインでは、静的リソースの管理、リ
ソース処理の為のマッパー、リソース提供用のサーブレットフィ
ルタ等を、わかりやすい DSL で管理可能な " フレームワーク " と
して提供しています。

まずは、Grails2.0 の BuildConfig.groovy を確認してみます。

plugins {

    compile ":hibernate:$grailsVersion"

    compile ":jquery:1.7.1"

    compile ":resources:1.1.5"

    build ":tomcat:$grailsVersion"

}

plugins の中でしっかりと Resources プラグインが定義されて

います。Resources プラグイン対応の jquery プラグインもインス
トールされています。

使用方法

■準備編
grails-app/views/layouts/main.gsp の head タグ内とコンテンツ

の最後に <r:layoutResources/> タグを挿入します。

<!doctype html>

            ...

    <head>

        <meta http-equiv="Content-Type" 

   content="text/html; charset=UTF-8">

        <meta http-equiv="X-UA-Compatible"

   content="IE=edge,chrome=1">

        <title><g:layoutTitle default="Grails"/></title>

            ...

        <r:layoutResources/>

        <g:layoutHead/>

    </head>

    <body>

            ...

        <r:layoutResources/>

        <g:javascript library="application"/>

    </body>

</html>

head 内とコンテンツの最後の部分へタグを挿入します。これ
で Resources プラグインを使用する準備は完了です。静的リソー
スはこの挿入したタグの部分に読み込まれる事になります。

■JavaScript を挿入
準備編で挿入した layoutResources タグ部分へ javascript を挿入

してみます。
grails-app/views/index.gsp の body 開始タグ直後に以下のコー

ドを書いて grails アプリケーションにブラウザからアクセスして
みましょう。

<r:script disposition="head">alert('コンテンツ前');</r:script>

<r:script disposition="defer">alert('コンテンツ後');</r:script>

検証 !　Grails2.0 の世界 Part.2

玉田  幸治（株式会社ニューキャスト）
基本的にきゃっきゃ♪ゆうとるひげのおっさん。

検証 ! Grails 1.4.0 の世界 Pt.22.0
祝 !! Grails 2.0.0 リリース

コンテンツの読み込み前と読み込み後に alert が表示されます。
html のソースコードを見てみましょう。

<!doctype html>

            ...

    <head>

            ...

        <script type="text/javascript">

  alert('コンテンツ前');

 </script>

            ...

    </head>

    <body>

            ...

        <div class="footer" role="contentinfo"></div>

        <div id="spinner" class="spinner"

  style="display:none;">

  Loading&hellip;

 </div>

        <script type="text/javascript">

  alert('コンテンツ後');

 </script>

    </body>

</html>

script タグの差し込み先が head タグ内とコンテンツの最後に
わかれて挿入されているのが確認できます。disposition="head"
で head 内の layoutResources へ挿入され、disposition="defer" で
コンテンツの最後の layoutResources へ挿入されます。
※ head タ グ 属 性 で 指 定 さ れ た も の は 最 初 に み つ か っ た

layoutResources タ グ 内 に 挿 入 さ れ ま す。head タ グ 内 に
layoutResources タグがなくてコンテンツの最後の箇所にしか
layoutResources タグがない場合はそちらに挿入されます。

css は disposition=head でヘッダー内の layoutResouces に読み
込みを行い、コンテンツ読み込み後に実行させたい javaScript や、
ページロードに影響を与える javaScript は disposition=defer でコ
ンテンツの最後に読み込むべきです。
※ <r:script> タグで定義された javaScript は require で読み込まれ

たライブラリより後に配備されます。必要なライブラリが先に
読み込まれている必要がある為にこのような動きになっていま
す。

コンテンツ後は読み込まれた後に実行されていることがわかり
ます。

■画像の挿入
web-app/images 内にある懐かしい画像を挿入してみましょ

う。

<r:img dir="images" file="grails_logo.jpg" />

以下のように変換され懐かしい画像が表示されます。

<img src="/あなたのアプリ/static/images/grails_logo.jpg" />

検証 !　Grails2.0 の世界 Part.2

設定ファイル
grails-app/conf 内 に リ ソ ー ス 定 義 フ ァ イ ル

（SomeNameResources.groovy という形で 'SomeName' に任意の
名前を付けた groovy ファイルを配置）、もしくは

grails-app/conf/Config.groovy に以下のように記述します。

grails.resources.modules = {

    module名 {

        dependsOn 'jquery' //依存がある場合
        defaultBundle 'defaultBundle' //バンドル名
        resource url:リソースファイルのパス,

  disposition: 'head', or 'defer'

    }

}

リソース定義ファイルは複数のファイルを使用することが可能
です。jquery プラグインなどはプラグインの中に既に Resources
プラグイン用の設定ファイルが含まれています。プラグインの中
にリソース定義ファイルを配置することで Resources プラグイン
対応のプラグインを作成することが可能です。

■モジュール
Resources プラグインではリソースのファイルをモジュールと

いう単位で管理しています。複数のリソースファイルをモジュー
ルという単位に統合し、gsp でモジュールを require タグで指定
することで利用が可能です。基本的なリソースを common モ
ジュールとしてまとめて、管理画面等の別のリソースを必要と
する場合、common モジュールを dependsOn した管理画面用の
admin モジュールを作るなどの手法が考えられます。

モジュール名称 { 

    dependsOn 'jquery' // 依存関係の定義
   設定してある場合は設定して
   あるモジュールが先に読み込
   まれます。
    defaultBundle 'monolith' // バンドル名称
    resource url: '/js/core.js', disposition: 'head'  

// モジュールのリソース
}

■バンドル
バンドルというセットを利用してリソースの上書きが可能にな

ります。リソースを上書きする場合、"overrides" を利用します。

overrides {

    jquery { defaultBundle '上書するためのバンドル' } 

} 

このように defaultBundle を別のバンドルで overrides すること
ができます。

たとえば、jQuery と他のライブラリをまとめてバンドルして、

上書きする場合は以下のようになります。

modules = { 

    core { 

        dependsOn 'jquery, utils'

        defaultBundle 'monolith'

        resource url: '/js/core.js', disposition: 'head' 

    }

    utils {

        dependsOn 'jquery'

        defaultBundle 'monolith'

        resource url: '/js/utils.js'

    }

    forms { 

        dependsOn 'core,utils'

        defaultBundle 'monolith'

        resource url: '/css/forms.css' resource url: '/js/forms.js' 

    }

    overrides {

        jquery { defaultBundle 'monolith' } 

    } 

}

■マッパー定義 
リソースフレームワークは " マッパー " を使用して、最終フォー

マットへリソースを変化させてユーザへ配信します。時にはこの
マッパーで処理させたくないリソースもあります。

特定のファイルを指定してマッパーの処理を行わないように定
義することも可能です。例えば、"hashandcache" マッパーの処
理をさせたくない場合は、

resource url: '/css/my.css', nohashandcache: true

上記のように、no マッパー名称 :true のように指定します。

また、ファイルタイプによって、マッパーの定義をする場合
は grails-app/conf/Config.groovy 内に以下のように指定が可能で
す。

grails.resources.zip.excludes = ['**/*.zip', '**/*.exe']

※ zip ファイルと exe ファイルを zip マッパーを通さない場合の設
定 (zip マッパーは zipped-resources で提供されています。)

grails.resources.bundle.excludes = ['**/*.less']

※ファイルタイプ less をバンドルしないようにする。

検証 !　Grails2.0 の世界 Part.2

開発中時のデバッグ指定
Resources プラグインにはデバッグの際に使用する機能が幾つ

か提供されています。開発環境では X-Grails-Resources-Original-
Src ヘッダーが追加されます。このヘッダーにはリソースを構成
した元のリソースが示されています。

また、URL に _debugResources=y のクエリパラメータを付加
することにより Resources プラグインは元のリソースファイルを
読み込むことが可能となります。また、ブラウザによるキャッシュ
を防ぐため、ユニークなタイムスタンプが付加されます。毎回ク
エリに _debugResources=y を付けなくてもいいように、

grails.resources.debug=true

と grails-app/conf/Config.groovy に記述することができます。
これにより、_debugResources=y のクエリパラメータを省いて
の開発が可能になります。開発中は development 環境内に追記
しておくことをおすすめします。

関連プラグイン

■zipped-resources プラグイン
ページロードの時間と帯域を短縮するためのプラグインです。

gif, jpeg, png の既に圧縮されているファイルを除いたコンテンツ
を自動的に圧縮するマッパーが用意されています。

grails install-plugin zipped-resources

このように、zipped-resources プラグインはインストールする
だけで使用できます。 

 

■cached-resources プラグイン
クライアントのブラウザにリソースをキャッシュさせる仕組み

のプラグインです。リソースのファイル名をそれぞれのハッシュ
値に変換し、キャッシングヘッダーを全てのリソースのレスポン
スにセットすることによりブラウザにキャッシュさせ、ページ
ロード時間の短縮を行います。

grails install-plugin cached-resources

同じく cached-resources プラグインはインストールするだけ
で使用できます。

まとめ
Grails 2.0 の世界として、駆け足で Resources プラグインの紹介

をしてきましたがこのプラグインを使用することで静的リソース
の管理が楽になることがおわかりいただけたでしょうか？まだま
だ紹介しきれていない部分もありますので公式のドキュメントを
一度みていただくことをおすすめします。それではまたお会いで
きる時まで ! 基本的にきゃっきゃゆうとるヒゲのおっさんがお送
りしました。

リンク
• tyama さんの jggug 発表スライド http://slidesha.re/n43zfq
• Resouces プラグインドキュメント  

http://grails-plugins.github.com/grails-resources/

コ ラ ム
jqueryプラグインの中の設定ファイルはどうなっているのか

jquery プラグインの中の resources プラグイン用のリソース
設定ファイル JQueryPluginResources.groovy を見てみます。

// Resource declarations for Resources plugin

def jqver = org.codehaus.groovy.grails.plugins.

jquery.JQueryConfig.SHIPPED_VERSION

modules = {

    'jquery' {

        resource id:'js', url:[plugin: 'jquery', 

 dir:'js/jquery', file:"jquery-${jqver}.min.js"], 

        disposition:'head', nominify: true

    }

    'jquery-dev' {

        resource id:'js', url:[plugin: 'jquery',  

 dir:'js/jquery', file:"jquery-${jqver}.js"], 

          disposition:'head'

    }

}

<r:require modules="jquery"/> を 利 用 す る こ と に よ り
圧 縮 さ れ た jquery-min を 読 み 込 む こ と が で き、<r:require 
modules="jquery-dev"/> で、圧縮前の jquery ライブラリを読
み込むことが可能になっているのがわかります。このように
開発環境とプロダクション環境をモジュールによって分けら
れるのも resources プラグインの利点といえます。

検証 !　Grails2.0 の世界 Part.2

http://slidesha.re/n43zfq
http://grails-plugins.github.com/grails-resources/

GDK48 投票アプリの作り方 ( 前編 )
今回は「Griffon 不定期便」をお休みして、JGGUG 合宿 2011 で

行われた GDK48 という企画で使われたアプリケーションついて
お話ししたいと思います。

GDK48 とは、みんなで G* なコードを書いて、それをみんなで
投票して 1 位を決めようという企画でした。企画の詳細な内容に
ついては合宿運営レンジャーレッドの資料を御覧ください。

http://www.slideshare.net/kazuchika/gdk48
GDK48 と言っておきながら、残念なことに 48 本までは集まり

ませんでしたが、大変役に立つコードが集まりました。合宿で使
用したアプリケーションは下記の URL で現在も見ることができ
ます。

http://gdk48.kiy0taka.org/gdk48/
今回はこの GDK48 総選挙システムの作り方についてお話しし

ていきます。GDK48 総選挙システムのソースコードは GitHub 上
で公開していますので、合わせて御覧ください。

http://git.io/gdk48

■GDK48 総選挙システム概要
GDK48 総選挙システムはとてもシンプルなものです。ストー

リーとしては次の 2 つになります。
• Gist に投稿した URL をツイートして GDK48 にエントリーする
• Twitter アカウントで認証済みのユーザが GDK48 に投票する

今回はこの「Gist に投稿した URL をツイートして GDK48 にエ
ントリーする」の実装についてお話ししていきます。

■アプリケーションの作成
GDK48 総選挙システムはシンプルなアプリケーションなので

Gaelyk で作るという選択肢もありましたが、Twitter 連携などの
必要な機能がプラグインで提供されているため Grails を使って実
装しました。Grails のバージョンは 1.3.7 を使いました。Grails2.0
が出たところですが、この記事でも 1.3.7 で作成していきます。
Grails2.0 については山本さんの記事を熟読してください。それで
は gdk48 という名前でアプリケーションを作成しましょう。

grails create-app gdk48

Gist に投稿した URL をツイートして GDK48
にエントリーする

GDK48 にエントリーする流れは次のような感じです。
1. GDK48 エントリー用のコードを書く
2. 書いたコードを Gist(http://gist.github.com) に貼り付ける
3. 貼りつけた Gist の URL とハッシュタグ「#gdk48」を付けて

ツイートする
例）「Groovy で Git コマンド。
　　https://gist.github.com/1341311 #gdk48」

ハッシュタグと Gist の URL があれば GDK48 総選挙システムが
勝手に拾ってくれてエントリーされます。

GDK48 エントリ用ドメインクラスの作成
それではエントリ用のドメインクラスを作成しましょう。エン

トリ用ドメインクラスと言っても、総選挙システムで管理する
のは Gist の ID だけにします。Gist の ID とは、Gist にコードを投稿
したときに割り当てられる URL の後ろにつく数字です。例えば、

「https://gist.github.com/1341311」 と い う URL な ら、Gist の ID
は「1341311」になります。コードの内容などはエントリー後に
修正されることもあるのでシステムではコード内容を保持せず、
Gist へのリンクが生成できれば良しとします。クラス名は Gist と
してドメインクラスを作成します。

cd gdk48

grails create-domain-class Gist

grails-app/domian/gdk48/Gist.groovy が生成されました。Gist
クラスには Gist の ID だけあればいいので、次のように編集しま
す。

package gdk48

class Gist {

    Long gistNo

    static constraints = {

        gistNo nullable:false, unique:true

    }

}

 JGGUG合宿2011企画アプリ
GDK48 投票アプリの作り方（前編）

07
ser ies 奥  清隆 （おく きよたか）

仕事でもときどき Groovy と戯れるプログラマ。
日本 Grails/Groovy ユーザーグループ関西支部長。
著書：『Seasar2 による Web アプリケーションスーパーサンプル』

GDK48 投票アプリの作り方（前編）

http://www.slideshare.net/kazuchika/gdk48
http://gdk48.kiy0taka.org/gdk48/
http://git.io/gdk48
http://gist.github.com
https://gist.github.com/1341311
https://gist.github.com/1341311

プロパティ gistNo が Gist の ID となります。gistNo は null 値を
許可せず、ユニークな値になるように制約を定義しておきます。

Gist コントローラの作成
ドメインクラスが出来たのでコントローラを作成しましょう。

grails create-controller Gist

grails-app/controllers/gdk48/GistController.groovy が生成され
ました。GDK48 総選挙システムでは、CRUD の画面は必要ないの
ですが、とりあえず動かしてみたい人はスカッフォルドで動きを
確認してみましょう。GistController.groovy を次のように編集し
ます。

package gdk48

class GistController {

    static scaffold = true

}

この状態でアプリケーションを起動すれば、スカッフォルドさ
れた CRUD の画面を動かせます。「grails run-app」コマンドを実
行してからブラウザで「http://localhost:8080/gdk48/gist」にア
クセスしてみましょう。最終的にこの CRUD の画面は必要ではな
いので詳細は省きますが、スカッフォルドでアプリケーションを
動かすことで、システムのイメージが掴めることもあるかと思い
ます。コントローラに少し工夫をするだけで CRUD の画面が確認
できるスカッフォルド機能はとても便利ですね。

それでは、実際に必要な処理を実装しましょう。GistController
に必要なアクションはエントリされた Gist の一覧を表示するため
のアクションです。一覧表示用に list という名前のクロージャを
定義します。

package gdk48

class GistController {

    def list = {

        def gists = Gist.list(max:5,

            offset:(params.offset?:'0').toLong(),

            sort:'gistNo',

            order:'desc')

        [gists:gists]

    }

}

list クロージャではエントリされた Gist の最大 5 件まで取得し
ます。リクエストパラメータ offset で開始件数を指定すればペー
ジングするようにします。検索結果は Gist の新しい順（gistNo の

昇順）でソートします。リクエストパラメータの入力チェックな
ど細かいことは気にしないことにします。

一覧画面の作成
Gist の一覧を表示する画面を作りましょう。grails-app/views/

gist/list.gsp ファイルを次のように作成します。

<%@ page import="gdk48.*" %>

<html>

<head>

  <meta name="layout" content="main" />

  <title>GDK48総選挙</title>

</head>

<body>

<div class="container">

  <h1>GDK48 総選挙</h1>

  <div class="paginateButtons">

    <g:paginate controller="gist" action="list" 

 total="${Gist.count()}" max="5" />

  </div>

  <div>

    <g:each var="gist" in="${gists}">

    <div class="gist">

      <script 

 src="https://gist.github.com/${gist.gistNo}.js">

      </script>

    </div>

    </g:each>

  </div>

  <div class="paginateButtons">

    <g:paginate controller="gist" action="list" 

 total="${Gist.count()}" max="5" />

  </div>

</div>

</body>

</html>

ページング用のタグと、script タグを使って Gist のコードを表
示 し て い ま す。Gist で は「https://gist.github.com/1341311.js」
のような URL で Gist コードを表示するための JavaScript コードを
取得することができます。

■テストデータの登録
この段階で一覧画面を確認しても、Gist を登録する部分を実

装していないのでデータは何も表示されません。しばらくは
BootStrap を使って適当な Gist を登録しておきましょう。grails-
app/conf/BootStrap.groovy を次のように編集します。

GDK48 投票アプリの作り方（前編）

http://localhost:8080/gdk48/gist
https://gist.github.com/1341311.js

import grails.util.Environment

import gdk48.Gist

class BootStrap {

    def init = { servletContext ->

        if (Environment.current == Environment.DEVELOPMENT) {

           [1261979, 1129161, 1059105, 1031115,

             1020384, 1020286, 1015558, 1012403].each {

             new Gist(gistNo:it).save(failOnError:true, 

     flush:true)

            }

        }

    }

    def destroy = {

    }

}

アプリケーション初期化のタイミングで、適当な Gist を何件
か登録する処理を追加しました。それでは「grails run-app」コ
マンドでアプリケーションを起動して「http://localhost:8080/
gdk48/gist/list」にアクセスしてみましょう。

次のような画面が表示されれば成功です。

■UI を改善する
Gist を一覧することが出来ましたが、見た目がすこし残念な

感 じ な の で、 少 し UI を 改 善 し ま し ょ う。Blueprint CSS(http://
blueprintcss.org/) という CSS フレームワークを使ってみます。
Grails には Blueprint CSS 用のプラグインがありますのでインス
トールします。

grails install-plugin blueprint

それでは Blueprint CSS を使って list.gsp を改善しましょう。

<%@ page import="gdk48.*" %>

<html>

<head>

  <meta name="layout" content="main" />

  <title>GDK48総選挙</title>

  <blueprint:resources/>

  <style>

  body {

    background-color: #FAEBD7;

  }

  .gist {

    margin-bottom: 30px;

  }

  </style>

</head>

<body>

<div class="container">

  <h1>GDK48 総選挙</h1>

  <div class="paginateButtons">

    <g:paginate controller="gist" action="list" 

 total="${Gist.count()}" max="5" />

  </div>

  <div class="span-26 last" 

 style="background-color: #FFB6C1; padding: 30px;"> 

    <g:each var="gist" in="${gists}">

    <div class="gist">

      <script src="https://gist.github.com/${gist.gistNo}.js">

        </script>

    </div>

    </g:each>

  </div>

  <div class="paginateButtons">

    <g:paginate controller="gist" action="list" 

 total="${Gist.count()}" max="5" />

  </div>

</div>

</body>

</html>

GDK48 投票アプリの作り方（前編）

http://localhost:8080/gdk48/gist/list
http://localhost:8080/gdk48/gist/list
http://blueprintcss.org/
http://blueprintcss.org/

これで次のような画面になりました。

まだ残念な感じがするかもしれませんが、これが私の限界です。Grails には他にも UI 系のプラグインがたくさんあるので、好きなも
のを使ってみるといいかと思います。

GDK48 エントリTweet を取得する
それでは、GDK48 エントリ Tweet を取得する処理を実装しま

しょう。処理の内容は次のとおりです。
1. 定期的にハッシュタグ「#gdk48」が含まれるツイートを

Twitter から検索
2. 検索結果のツイートに Gist の URL が含まれていればデータを

登録

■Job の作成
定期的に処理を実行するために Quartz プラグインを使用しま

す。Quartz プラグインは、スケジューリングされた処理を実行す
るための Java ライブラリである Quartz(http://quartz-scheduler.
org/) の Grails プラグインです。

まず、Quartz プラグインをインストールしましょう。

grails install-plugin quartz

Quartz プラグインをインストールすると「create-job」コマン
ドを使って定期的に行う処理を実装する Job クラスを作成できま
す。それでは GDK48エントリツイート取得用 Job を作成しましょ
う。

grails create-job TweetCrawler

こ れ で、grails-app/jobs/gdk48/TweetCrawlerJob.groovy と い
うファイルが生成されました。TweetCrawlerJob は少しコードが
長いので、紙面の都合上ポイントだけ解説しておきます。完全
なソースはサンプルコード (http://git.io/gdk48job) を御覧くださ
い。

■トリガーの定義
今回は 1 分おきにこの Job を実行するようトリガーを定義しま

す。Quartz ではクーロン式を使ったトリガーなどいくつかの形式
でトリガーを定義できるようになっています。ここでは「simple」
タイプのトリガーを定義します。

static triggers = {

    simple name: 'TweetCrawlerTrigger', 

    startDelay: 10000, repeatInterval: 60000

}

name 属性は任意のトリガー名、startDelay はアプリケーショ
ンが起動してから初めてトリガーが起動されれるまでの時間をミ
リ秒で指定、repeatInterval はトリガーを起動する間隔をミリ秒
で指定します。

この設定では、「TweetCrawlerTrigger」という名前でアプリケー
ションが起動してから 10 秒後に最初の Job が実行され、その後 1
分間隔で Job が実行されます。

GDK48 投票アプリの作り方（前編）

http://quartz-scheduler.org/

■Job の並列実行
Job は 1 分おきに実行されますが、1 分以上処理に時間がかかっ

た場合、2 つの Job が平行して実行されます。TweetCrawlerJob
は並列に実行されても問題はありませんが、意味が無いので並列
に実行されないようにしておきます。TweetCrawlerJob にプロパ
ティ concurrent を定義しておけば、並列実行を制御することが
できます。

def concurrent = false

プロパティ concurrent を定義しなかったり、値を true にした
場合は並列に実行されることを許可します。

■テスト実行時のトリガーの制御
今回はテストコードについては紹介しませんが、サンプルコー

ドにはいくつかテストコードを書いています。デフォルトの設定
ではテスト実行時にもトリガーが起動されてしまうので、予期し
ないタイミングで実行されテストが失敗するケースも出てきてし
まいます。テスト時にトリガーを起動しないように次のような
ファイルを grails-app/conf/QuartzConcifg.groovy として作成し
ておきます。

quartz {

    autoStartup = true

    jdbcStore = false

    waitForJobsToCompleteOnShutdown = true

}

environments {

    test {

        quartz {

            autoStartup = false

        }

    }

}

■Twitter4J のライブラリを追加
Twitter か ら ハ ッ シ ュ タ グ「#gdk48」 で 検 索 す る た め に

Twitter4J を使用します。grails-app/config/BuildConfig.groovy に
Tiwtter4J の依存関係を追記しておきましょう。

repositories {

    grailsPlugins()

    grailsHome()

    grailsCentral()

    mavenLocal()

    mavenCentral()

}

dependencies {

    compile 'org.twitter4j:twitter4j-core:jar:2.2.5'

}

■処理の実装
TweetCrawlerJob のメソッド execute() を実装します。

def execute() {

    def twitter = new TwitterFactory().instance

    twitter.search(new Query('#gdk48'))

   .tweets.each { parse it }

}

Twitter4J の検索 API を使ってハッシュタグで検索し、解析して
データを登録するためのメソッド parse() に検索結果を渡します。
このまま実行してもおそらく「#gdk48」では検索結果が返って
こないので、適当に変更して実行してみてください。

メソッド parse() については詳細な解説を省きますが、ツイー
トの本文に Gist の URL が含まれるかどうかを判定してデータを登
録するように実装します。この処理のポイントとしては次の 2 点
が考えられます。
• URL は短縮 URL(http://t.co/BOFnltIQ のような形式など ) の場合

がある
• 短縮 URL は他の短縮 URL サービスで短縮された URL をさらに

短縮している場合がある
この 2 点を考慮して、ツイート内容から Gist の URL を取得する

ために、ツイート内の URL でリクエストを試みてから、短縮 URL
の場合に返される Location ヘッダを取得し判定するようにしま
す。処理の流れは次の図のようになります。

GDK48 投票アプリの作り方（前編）

http://t.co/BOFnltIQ


G*Magazine vol.4

13

Location ヘッダの取得には java.net.URL クラスを使って次のよ
うにヘッダだけ取得しています。

def conn = new URL(url).openConnection()

conn.followRedirects = false

def location = conn.getHeaderField('Location')

まとめ
今回は、GDK48にエントリされた Gist を登録し、アプリケーショ

ンで一覧するところまでを紹介しました。本当は 1 回ですべてを
紹介する予定でしたが、文章にすると予想以上にボリュームが
あったので、続きは次回に回したいと思います。次回は Twitter
を使った認証と投票の仕組みについてです。それではまた次回お
会いしましょう！

GDK48 投票アプリの作り方（前編）



G*Magazine vol.4

14

今年 10 月に開催された JavaOne 2011 で、Java の次世代クライ
アントソリューションである JavaFX 2.0 がついに正式リリースさ
れました。そして G* エコシステムにも、この JavaFX を Groovy
に使いこなすための GroovyFX という新しい仲間が増えました。

このシリーズでは、この GroovyFX を取り上げ、その使い方や
魅力を紹介していきたいと思います。

JavaFX って何だっけ？
JavaFX は、Java SE プラットフォームに、従来 Java があまり得

意でなかった下記のような機能を追加する拡張ライブラリです：

• アニメーションやエフェクト、3D といった高度なグラフィッ
クス

• メディア（動画や音声）の再生と制御
• web ブラウザとの統合・連携

Java SE を拡張する JavaFX

JavaFX 2.0 は、今日の GPU の性能をフルに引き出せるよう設計
された新しい描画エンジンを備え、シーングラフという形で表
現されたグラフィックス /UI 要素を高速に描画します。シーング
ラフの各ノードには多彩なエフェクトやトランジションを適用
することができます。さらに、WebKit ベースのブラウザ実装や、
GStreamer ベースのメディアプレーヤを内蔵し、これらをシーン
グラフに埋め込んだり、Java から自在に制御することもできま
す。

JavaFX の開発が始まったのは Oracle による Sun 買収の数年前
までさかのぼり、途中何度かの開発方針の見直しなどを経て、現
在の形に落ち着きました。現時点では、Windows 用の正式版と、
Mac OS X 用の開発者プレビュー版がリリースされており、2012
年には Linux 版も予定されています。

今 年 の JavaOne で は、 オ ー プ ン ソ ー ス 化 と JCP(Java 
Community Process) での標準化の計画も発表され、さらに、今
後の JDK7 リリースにバンドルされるようになること、そして次
期メジャーアップデートである JDK8 に標準 API として統合され
ることなども明らかにされました。

これにより、JavaFX はもはや単なる拡張ライブラリの一つで
はなく、AWT や Swing の後継として、今後の Java のクライアン
ト UI 開発の主役の座につくことが確実になったのです。

GroovyFX 登場！
GroovyFX は、Jim Clarke (Oracle) と Dean Iverson (Virginia 

Tech) によって開発された、Groovy による JavaFX のラッパーで
す。Groovy 活用法７パターンでいうところの、"Lipstick" パター
ンに該当します。（「Java ライブラリをラップして使いやすく」）

JavaFX 2.0 の API は す べ て Java ベ ー ス な の で、Groovy か ら
普通の Java ライブラリのように利用することも可能ですが、
GroovyFX では、さらに次のような機能を提供することによって、
JavaFX をより Groovy らしく、簡単・便利に利用できるようにし
ています：

• Groovy のビルダーを活用した DSL
• SceneGraphBuilder: シーングラフの構築
• TimelineBuilder: タイムラインの構築

• Groovy の言語機能を活用した便利な記法
• Duration リテラル : 100.ms, 5.s, 3.h など
• 疑似定数 : red, vertical, ease_out など
• 属性値の自動型変換 : 文字列→ Font、リスト→ Point3D など

JavaFX 1.x までは JavaFX Script という独自の言語が使われてお
り、この言語によって、JavaFX のシーングラフ構築などを簡潔
かつ見通しよく記述することができました。しかし、JavaFX 2.0
ではこの言語が廃止され、すべてが Java ベースになったため、
記述は煩雑になってしまいました。JavaFX Script の後継として
GroovyFX を使えば、Groovy が持つ豊かな言語機能と、GroovyFX
が提供する JavaFX に特化したサポートの組み合わせが、JavaFX 
Script の抜けた穴を補って余りあるでしょう。

■入手とビルド
残念ながら GroovyFX はまだアルファリリースの段階です。zip

や jar などの便利なバイナリディストリビューションもありませ

GroovyFX で遊ぼう！（導入編）

08
ser ies 関谷 和愛（せきや かずちか）

Groovy、Java、Apple 製品を愛するおっさんエンジニア。日本 Grails/Groovy ユーザーグループ運営委員長。
共著『プログラミング GROOVY』、共訳『Groovy イン・アクション』

GroovyFX で遊ぼう！（導入編）

ん。利用するためにはまずソースを入手し、自分でビルドする必
要があります。

とは言ってもビルドには難しいところはありませんので、下記
の手順にそってぜひやってみてください：

1. 必要なソフトウェアの用意
• JavaFX SDK ( 最新バージョン : 2.0.2)

http://www.oracle.com/technetwork/java/javafx/downloads/index.html
JavaFX のランタイム。Java SE 7 Update 2 からは、Java SE に

もバンドルされています。インストール後、環境変数 JAVAFX_
HOME にインストール先を設定しておいてください。

• Groovy ( 最新バージョン : 1.8.4)
http://groovy.codehaus.org/Download
Groovy のランタイム。（G* Magazine の読者はインストール済

みだと思いますが！）

• Gradle ( 最新バージョン : 1.0-milestone-6)
http://www.gradle.org/downloads
Groovy で作られた汎用ビルドツール。

2. ソースを入手
リポジトリからソースをチェックアウトします。

% svn co http://svn.codehaus.org/gmod/groovyfx/trunk/

執筆時点のリビジョンは 1072 でした。

3.ビルド実行
build.gradle のあるディレクトリで gradle コマンドを実行しま

す。

% cd groovyfx/trunk/groovyfx/

% gradle build

4. クラスパスの設定
build/libs の下にできた groovyfx-0.1-SNAPSHOT.jar にクラスパ

スを設定します。もしまだなら、$JAVAFX_HOME/rt/lib/jfxrt.jar
もクラスパスに入れておきます。

以 上 で、GroovyFX の 利 用 準 備 が 整 い ま し た。src/demo/
groovy の下にたくさんのデモスクリプトがあるので、実行して
みたりソースを眺めてみるとよいでしょう。

簡単なプログラムの例
では実際に GroovyFX プログラムの例を見てみましょう。

import groovyx.javafx.*

key = '※Google+のAPIキー'

uid = 110611905999186598367  // user ID

url = "https://www.googleapis.com/plus/v1/

people/$uid?key=$key".toURL()

json = new groovy.json.JsonSlurper().

parseText(url.text)

GroovyFX.start {

  def sg = new SceneGraphBuilder()

  stage = sg.stage(title: "Profile", width: 640, height:380) {

    scene(fill: black) {

      imageView(x: 20, y: 40, rotationAxis: [0, 1.0, 0]) {

        image(json.image.url.replaceAll(/sz=50/, 'sz=200'))

        effect reflection(fraction: 0.25)

        transition = 

          rotateTransition(1.s, from:0, to:360, tween: ease_out)

        onMouseClicked { transition.play() }

      }

      text(x: 240, y: 60, text: json.displayName, 

        fill: white, font: "32pt", textOrigin: "top") {

        effect bloom()

      }

      text(x: 240, y: 120, text: json.tagline, 

        fill: white, font: "16pt", textOrigin: "top")

    }

  }

  stage.show()

  transition.play()

}

※下記から Google+ の API キーを入手し、key 変数に設定してく
ださい。
https://code.google.com/apis/console/

GroovyFX で遊ぼう！（導入編）

http://www.oracle.com/technetwork/java/javafx/downloads/index.html

このプログラムは、Google+ の API を使って特定のユーザのプ
ロフィール情報を取得し、次のような画面を出力します：

実行結果

簡単なアニメーション要素として、起動時に一度、そしてクリッ
クするたびにプロフィールイメージが回転するようにしてありま
す。

■コードの概要説明
最初の数行では、Google+ の API にアクセスして、JSON のレ

スポンスを受け取っています。JavaFX には JSON パーサが盛り
込まれる予定もあったのですが、最終的には入りませんでした。
Groovy には標準で JSON パーサが装備されており、このような
web API の利用では大いに役立ちます。

GroovyFX.start {

  def sg = new SceneGraphBuilder()

この部分は GroovyFX アプリの決まり文句です。

GroovyFX クラスは JavaFX の Application クラスのサブクラス
で、start メソッドも Application クラスのものと同様の意味を持
ちます。ただし、GroovyFX ではクロージャを渡せるようになっ
ており、このクロージャが JavaFX のアプリケーションスレッド
で実行されます。シーングラフの構築などはこのクロージャの中
で行われる必要があります。

次の行では SceneGraphBuilder のインスタンスを生成していま
す。以下、このビルダーを使ってアプリケーションのシーングラ
フを構築しています。

このサンプルのシーングラフは下図のようになっています：

サンプルのシーングラフ構造

トップレベルのウィンドウに相当する Stage の下に、各ノー
ドを収容するコンテナである Scene、さらにその下 に 1 つの
ImageView と 2 つの Text ノードがあります。各コンテナやノード
の属性は、Groovy の他のビルダーと同様、マップのリテラル形
式（key:value）で記述され、エフェクトやトランジション、イ
ベントハンドラなども各ノードの子要素として記述されていま
す。

上記の構造図と比較しながらコードを眺めると、シーングラフ
の構造と各ノードの属性や振る舞いが、簡潔かつ明快に表現され
ていることがよくわかると思います。

もう少し細かいところも見てみましょう。

rotateTransition(1.s, from:0, to:360, tween: ease_out)

この行には 1.s という表現がありますが、ここは本来 JavaFX で
は Duration を指定するところです。GroovyFX では、Groovy のメ
タプログラミング機能を使って Number クラスに getS、getMs な
どのメソッドを追加して、Number から Duration への変換を実現
しています。これにより、5.s、100.ms、3.h などの表現が可能に
なっています。

また、上のコードでの ease_out や、色を指定している black、
white などはいったい何者でしょうか？これらは一見定数のよ
うに見えますが、実はそうではありません。SceneGraphBuilder
クラスの中に定義した propertyMissing メソッドで未知のプロパ
ティへのアクセスをフックし、対応する値に変換するというから
くりで、簡潔な表現の疑似定数として実現されています。

さらに、JavaFX なら Font や Point3D、VPos といった型で指定
すべき属性に対して、リストや文字列などが使われていることに
も気づかれたかもしれません。これらは FXHelper というクラス
で、渡された型に応じて、しかるべきクラスに自動変換する仕組
みが実装されています。

以上のように、GroovyFX では、ビルダーによるシーングラフ
の宣言的な構造記述に加え、Groovy の機能を駆使して記述を簡
潔にするさまざまな仕組みによって、JavaFX の快適な利用を実
現しているのです。

GroovyFX で遊ぼう！（導入編）

まとめ
今回は GroovyFX の概要とインストール、そしてごく簡単な

コードを例にその特徴を見てきました。

GroovyFX( と JavaFX) に は、 変 数 の バ イ ン ド 機 構 や FXML、
Swing や Griffon との統合、HTML5 コンテンツとの連携など、面
白い話題がまだまだたくさんあります。次回以降はそういった話
題を取り上げ、サンプルコードを使ってみなさんと一緒に「遊ん
で」いきたいと思います。どうぞお楽しみに！

■リンク
• JavaFX http://javafx.com/
• GroovyFX http://groovy.codehaus.org/GroovyFX
• 使用したサンプルコード https://gist.github.com/1341142

GroovyFX で遊ぼう！（導入編）

http://javafx.com/
http://groovy.codehaus.org/GroovyFX
https://gist.github.com/1341142

JGGUG でサポートスタッフを担当させていただいている @
nobusue と申します。

今回は Groovy ベースのテスト支援ツールである Betamax の紹
介記事を書かせていただくことになりました。G*Magazine には
初登場となりますが、よろしくお願いいたします。

Betamax とは
Betamax は Web へのアクセスを記録して再生することのでき

る record/playback proxy です。
最 近 の ア プ リ ケ ー シ ョ ン は Twitter や Facebook な ど 外 部 の

WebAPI と連携するものが多くなってきていますが、Betamax を
利用すると実際に WebAPI や Web サイトへのアクセスを行わず
にアプリケーションのテストを行うことができます。

Betamax は HTTP リクエスト・レスポンスのペアを、HTTP リ
クエストの内容をキーとして「tape」というテキストファイル
(YAML) に記録します。テストケースに対して tape を指定してや
ることで、既に記録済みの HTTP リクエストに対しては tape の内
容が再生されます。未記録の HTTP リクエストであれば、実際に
得られたレスポンスをキャプチャして tape に記録します。

Betamax を使うことで、例えば以下のような効果が期待できま
す。
• WebAPI を利用するアプリケーションのテストをオフラインで

行うことができる。
新幹線で移動中でも OK。

• いちいち WebAPI にアクセスする必要がなくなるため、スロー
テスト問題を解決できる。

• WebAPI 側の障害や仕様変更などからテストを分離できるた
め、テストを安定化することができる。

• 再現が難しいレアケースなどに対するテストを容易にできる。
例えば Twitter API の呼び出し回数制限に引っかかった場合な
ど。

• tape を資産として再利用したり、加工したりできる。
Betamax の発想の元となった vcr(Ruby ベース ) というプロダク
トでは、既に多くの資産が蓄積されています。

Betamax の開発者は Rob Fletcher 氏です。Rob は Geb のコミッ
タでもあり、Betamax には Geb/Spock/Grails といった G* エコシ
ステムとの連携があらかじめ織り込まれています。

Hello, Betamax
何はともあれ、まず実際に Betamax を動かしてみましょう。
http://grails.org/ にアクセスして、HTTP レスポンスのステータ

スコードが 200 であることを確認するテストケースを Spock で記
述してみます。テストケースは以下のようになります。

▼src/test/groovy/MySpec.groovy

import betamax.Betamax

import betamax.Recorder

import spock.lang.*

import org.junit.*

import groovyx.net.http.RESTClient

import org.apache.http.impl.conn.

ProxySelectorRoutePlanner

class MySpec extends Specification {

  @Rule Recorder recorder = new Recorder()   

// (1)Recorderの設定

  @Shared RESTClient http = new RESTClient()

  def setupSpec() {

    http.client.routePlanner = 

 new ProxySelectorRoutePlanner(   

// (2)Proxyの設定
         http.client.connectionManager.schemeRegistry, 

   ProxySelector.default)

  }

  @Betamax(tape = "my tape")   // (3)Tapeの設定
  def "simple http get response data"(){

    when:

      def response = http.get(uri:"http://grails.org/")

    then:

      response.status == 200

  }

}

(1) では、JUnit の @Rule アノテーションを利用して、Betamax
の Recorder を MethodRule として定義しています。この設定によ
り、テストケースで Betamax の機能が利用できるようになりま
す。

(2) では、RESTClient の proxy 設定として JVM の proxy 設定が反
映されるようにしています。（Betamax は JVM の proxy 設定を利
用してリクエスト・レスポンスのキャプチャーを行います。）

(3) では、テストケースを実行する際にリクエスト・レスポ

Betamax ことはじめ

ンスを記録する tape を指定しています。ここで指定した名称が
tape のファイル名として利用されます。

このテストケースをビルドして実行するための Gradle ビルド
ファイルは次のようになります。

▼build.gradle

apply plugin:"groovy"

repositories {

  mavenCentral()

}

dependencies {

  groovy "org.codehaus.groovy:groovy-all:1.8.4"

  testCompile "com.github.robfletcher:betamax:1.0"

  testCompile("org.codehaus.groovy.modules.

       http-builder:http-builder:0.5.1") {

    exclude module:"groovy"

    exclude module:"httpclient"

  }

  testCompile("org.spockframework:spock-core:0.5-groovy-1.8") {

    exclude mogule:"groovy-all"

  }

}

（ こ れ ら の コ ー ド は https://github.com/nobusue/betamax-
examples/tree/master/HelloBetamax か ら 入 手 で き ま す。 こ ち
らは Gradle を導入していない環境でも実行できるように Gradle 
Wrapper を仕込んであります。）

それではテストケースを実行してみましょう。build.gradle の
配置されているディレクトリから以下を実行します。

> gradle test

(Gradle Wrapperを使う場合) > gradlew test

「BUILD SUCCESSFUL」と表示されれば、テストは成功です。
こ の と き、src/test/resources/betamax/tapes を 確 認 す る と、

my_tape.yaml というファイルが生成されているはずです。これ
が Betamax が記録した HTTP リクエスト・レスポンスです。

▼my_tape.yaml( 抜粋 )

!tape

name: my tape

interactions:

- recorded: 2011-12-17T10:07:58.216Z

  request:

    method: GET

    uri: http://grails.org/

    headers:

      Accept: '*/*'

      Accept-Encoding: gzip,deflate

      Host: grails.org

      Proxy-Connection: Keep-Alive

  response:

    status: 200

    headers:

      Content-Encoding: gzip

      Content-Language: en-US

      Content-Type: text/html;charset=UTF-8

      Date: Sat, 17 Dec 2011 10:07:50 GMT

      Server: Apache/2.2.3 (Red Hat)

      Vary: Accept-Encoding

    body: "<!DOCTYPE html>\n<!--[if lt IE 7 ]> 

  <html class=\"ie6\"> <![endif]-->

  \n<!--[if IE 7 ]>

・・・

HTTP リクエストの記録日時、URI とヘッダ、およびレスポン
スのヘッダとボディが記録されていることがわかります。

それでは、tape が記録済みの状態で再度テストを実行してみ
ましょう。（初回実行時は Gradle が依存ライブラリを取得するの
に時間がかかるため、何回かテストを実行してみるとよいです。）
記録済みの URI に対するリクエストを Betamax が横取りして tape
を再生するため、tape なしでテストを実行した場合よりも早く
テストが完了するはずです。私の環境では、tape なしの場合は
ビルド完了まで 10 秒程度かかっていましたが、tape ありの場合
は 6 秒程度まで短縮できました。

また、my_tape.yaml をリネームして保存しておき、再度 tape
を記録して比較してみると、レスポンスの差分を確認すること
ができます。grails.org のコンテンツが変更されていなければ、
recorded と Date ヘッダの値のみが変わっていることが確認でき
るはずです。

Grails + Geb + Betamax サンプルコードの実行
も う 少 し 実 用 的 な 使 い 方 と し て、Grails の テ ス ト ケ ー ス で

Betamax を使う場合にどうすればよいかを見てみましょう。
Betamax の ソ ー ス を 見 て み る と、https://github.com/

robfletcher/betamax/tree/master/examples/grails-betamax 以 下

Betamax ことはじめ

にサンプルの Grails アプリケーションがあります。これは Twitter 
API 経由で「betamax」というキーワードを含む tweet を表示す
るアプリケーションですが、テストケースの中で Betamax を利
用して実際の API 呼び出しをエミュレートするようになっていま
す。

ただし、2012/12/18 時点では、Twitter Web サイトの仕様変更

によりオリジナルの Betamax のサンプルではテストケースが通
らなくなってしまっています。Twitter の仕様変更に対応した修
正を施したバージョンを https://github.com/nobusue/betamax/
tree/fix-grails-sample で公開していますので、「テストが fail する
のは気持ち悪い」という方はこちらをご利用ください。参考まで
に、オリジナルからの修正点は以下のとおりです。

diff --git a/examples/grails-betamax/grails-app/views/twitter/tweets.gsp
 b/examples/grails-betamax/grails-app/views/twitter/tweets.gsp
index 7f84680..187706d 100644
--- a/examples/grails-betamax/grails-app/views/twitter/tweets.gsp
+++ b/examples/grails-betamax/grails-app/views/twitter/tweets.gsp
@@ -4,7 +4,7 @@
   <li>
    <blockquote>
     <p>${it.text}</p>
-    <small><a href="http://twitter.com/${it.user}" rel="external">@${it.user}</a></small>
+    <small><a href="http://m.twitter.com/${it.user}" rel="external">@${it.user}</a></small>
    </blockquote>
   </li>
  </g:each>

diff --git a/examples/grails-betamax/test/functional/betamax/examples/TwitterPageSpec.groovy
 b/examples/grails-betamax/test/functional/betamax/examples/TwitterPageSpec.groovy
index 53e03d7..340a499 100644
--- a/examples/grails-betamax/test/functional/betamax/examples/TwitterPageSpec.groovy
+++ b/examples/grails-betamax/test/functional/betamax/examples/TwitterPageSpec.groovy
@@ -50,7 +51,7 @@ class TwitterPageSpec extends GebSpec {
   $('#tweets li').eq(0).find('small a').click()
 
   then:
-  title == "Christine Romero (@la_dyosa) on Twitter"
+  $('div.user-screen-name').text() == "la_dyosa (Christine Romero)"
  }
 
 }

サンプルアプリケーションの実行には Grails が必要ですので、
別途導入しておいてください。( 私の環境では Grails-1.3.7 で動作
確認を行いました。)

実 行 方 法 は 簡 単 で、 ソ ー ス を clone し て examples/grails-
betamax/ に移動し、以下を実行するだけです。

> grails run-app

「Server running. Browse to http://localhost:8080/grails-
betamax」が出力されたら、ブラウザでアクセスすると以下の画
面が表示されます。

「twitter」リンクをクリックすると、Twitter から「betamax」
を含む tweet を 10 件と、Twitter クライアントの分布を表示する
画面に遷移します。

このアプリケーションは画面表示毎に Twitter API を実行して
いるので、ためしに Twitter で「betamax 最高！」などとつぶや
いてから、画面を再表示してみてください。あなたの tweet が一
覧に表示されるはずです。

アプリケーションの動作を確認したら、サーバーを停止してお
いてください。

次に、テストを実行してみます。このサンプルではユニッ
ト テ ス ト (test/unit/betamax/examples/) と 機 能 テ ス ト (test/
functional/betamax/example) が提供されています。さきほどと
同様に以下を実行してみてください。

Betamax ことはじめ

> grails test-app

テストコードがビルドされ、ユニットテストおよび機能テスト
が実行されます。機能テスト実行時は自動的にアプリケーション
が起動されるため、テスト完了まで少し時間がかかります。

Tests PASSED - view reports in target\test-reports

Application context shutting down...

Application context shutdown.

という表示が出たらテストは完了です。taget/test-reports/
html/index.html を開いてテスト結果を確認してみてください。

テストが完了したところで、具体的にテストケースで何をして
いるのか確認してみましょう。

▼test/unit/betamax/examples/TwitterServiceSpec.groovy

class TwitterServiceSpec extends UnitSpec {

  File baseDir = BuildSettingsHolder.settings?.baseDir ?: new File("examples/grails-betamax")
  @Rule Recorder recorder = new Recorder(tapeRoot: new File(baseDir, "test/resources/tapes"))

  TwitterService service = new TwitterService()

  def setupSpec() {
    def log = Logger.getLogger("betamax")
      log.addHandler(new ConsoleHandler())
    }

  def setup() {
    def restClient = new RESTClient()
    restClient.client.routePlanner = new ProxySelectorRoutePlanner(
      restClient.client.connectionManager.schemeRegistry, ProxySelector.default)
    service.restClient = restClient
  }

  @Betamax(tape = "twitter success")
  def "returns aggregated twitter client stats when a successful response is received"() {
    when:
      def clients = service.tweetsByClient("betamax")

    then:   // (1)Twitterクライアントの分布のテスト
      clients.size() == 6
      clients["\u00DCberSocial for BlackBerry"] == 4
      clients["TweetDeck"] == 2
      clients["Echofon"] == 1
      clients["Mobile Web"] == 1
      clients["Snaptu"] == 1
      clients["Twitter for BlackBerry\u00AE"] == 1
  }

  @Betamax(tape = "twitter success")
  def "only retrieves tweets containing the search term"() {
    when:
      def tweets = service.tweets("betamax")

    then:   // (2)betamaxを含むtweetの検索結果件数と内容のテスト
      tweets.size() == 10
      tweets.every { it.text =~ /(?i)betamax/ }
  }

  @Betamax(tape = "twitter rate limit")
  def "sets an error status when twitter rate limit is exceeded"() {
    when:
      service.tweetsByClient("betamax")

    then:   // (3)異常系(例外発生)のテスト
      thrown TwitterException
  }
}

Betamax ことはじめ

(1) と (2) はいずれも Twitter API の実行結果に依存しているの
で、本来であればテスト用にモックやスタブを用意しなければな
りません。しかし、ここでは Betamax の tape を利用することで、
Twitter API の呼び出しを Betamax の proxy で代替しています。

(3) についても同様で、Twitter API の呼び出し回数制限に引っ
かかったケースを tape として記録しておき、それを再生するこ
とでテストを実行しています。このようなテストは WebAPI に負
荷をかけるなどの問題があるため通常は再現が難しく、モックや
スタブで置き換えるのも一筋縄ではいきません。Betamax が効果
を発揮する典型的なケースであるといえます。

Betamax をソースからビルドする場合の注
意点

■SonaType のアカウント設定
Betamax のソースコードを clone してビルドしようとすると、

「No such property: sonatypeUsername」 と い う 例 外 が 発 生 し
てビルドができません。これは「sonatypeUsername」および

「sonatypePassword」が設定されていないことが原因です。
これらの設定は SonaType に成果物 ( アーティファクト ) をアッ

プロードする際に必要となるもので、単にビルドして利用する
だけなら利用されることはありません。回避策として、build.
gradle と同じディレクトリに以下のファイルを作成してくださ
い。

▼gradle.properties

sonatypeUsername=

sonatypePassword=

■一部のテストが通らない
Betamax の実装はエンコーディングやタイムゾーンの考慮が不

十分な箇所があり、日本語環境で実行すると通らないテストがあ
ります。単に使うだけであれば特に実害はないので無視していた
だいてかまいません。

我こそはと思われる方は、ぜひバグ取りにご協力ください。

まとめ
今 回 は、Groovy で 実 装 さ れ た record/playback proxy で あ る

Betamax についての簡単な紹介と、Grails のテストケースでの活
用について説明しました。次回は Betamax のコンフィグレーショ
ンや、内部動作についてご紹介したいと思います。

Betamax の最新安定バージョンは 1.0 ですが、まだまだ発展途
上のプロダクトであり 2012 年 12 月現在も活発な開発が続いてい
ます。Betamax について興味をもたれた方は、ぜひ以下を参照し
てみてください。バグフィックスや機能拡張への貢献も歓迎です。
• Betamax Web サイト

http://robfletcher.github.com/betamax/
• Betamax ソースコードリポジトリ

https://github.com/robfletcher/betamax

Betamax ことはじめ

はじめに
今回ご紹介する Grails プラグインは、RabbitMQ プラグインで

す。
本記事は、次の環境で動作確認をしております。

• OS : Mac OS X 10.6.8
• Java : 1.6.0_29
• Grails : 1.3.7

RabbitMQとは
RabbitMQ とは、Advanced Message Queuing Protocol(AMQP)

を使用したオープンソースのメッセージングシステムです。
2010 年に Spring Source 社が買収し、現在は同社が RabbitMQ の
開発・サポートを行っています。

詳しくは、RabbitMQ を参照してください。

RabbitMQ プラグインとは
RabbitMQ プラグインとは、RabbitMQ との統合・連携をする

ためのプラグインです。RabbitMQ プラグインは、メッセージ送
受信を高レベルに抽象化するため、Spring AMQP を使用してい
ます。

RabbitMQ プラグインのインストール
アプリケーションを作成し、RabbitMQ プラグインをインス

トールします。

$ grails create-app rabbit

$ cd rabbit

$ grails install-plugin rabbitmq

RabbitMQ サーバの設定
RabbitMQ を利用するための設定は、grails-app/conf/Config.

groovy に記述します。

rabbitmq {

    connectionfactory {

        username = 'guest'

        password = 'guest'

        hostname = 'localhost'

    }

}

設定できるプロパティは、次のとおりです。

プロパティ名 説明 デフォルト値

rabbitmq.
connectionfactory.
username

RabbitMQ サーバ
に接続するための
ユーザ名

なし

rabbitmq.
connectionfactory.
password

RabbitMQ サーバに
接続するためのア
カウント用のパス
ワード

なし

rabbitmq.
connectionfactory.
hostname

RabbitMQ サーバの
ホスト名

なし

rabbitmq.
connectionfactory.
virtualHost

RabbitMQ サーバに
接続するための仮
想ホスト名

'/'

rabbitmq.
connectionfactory.
channelCacheSize

接続チャネルの
キャッシュサイズ

10

rabbitmq.
concurrentConsumers

メッセージハンド
ラごとに作成する
同時コンシューマ
の数

1

Grails Plugin 探訪
第 5 回

~ RabbitMQ プラグイン ~ 
URL: http://www.grails.org/plugin/rabbitmq

プラグインのバージョン : 0.3.2
対応する Grails のバージョン : 1.2 以上

杉浦孝博

最近は Grails を使用したシステムの保守をしている自称プログラマ。
日本 Grails/Groovy ユーザーグループ事務局長。
共著『Grails 徹底入門』、共訳『Groovy イン・アクション』

Grails Plugin 探訪　第 5 回〜 RabbitMQ プラグイン〜

キューの設定
キューの設定は、次の 2 つの方法があります。

• Spring Bean で定義する
• DSL で定義する

■Spring Bean で定義する
org.springframework.amqp.core.Queue 型 の Spring Bean の 定

義を、grails-app/conf/spring/resources.groovy にします。
例えば、次のようになります。

beans = {

    myQueue(org.springframework.amqp.core.Queue, 'myQueueName') 

    myOtherQueue(org.springframework.amqp.core.Queue, 

   'myOtherQueueName') {

        autoDelete = false

        durable = true

        exclusive = false

        arguments = [arg1: 'val1', arg2: 'val2']

    }

}

■DSL で定義する
先 程 の Spring Bean の 定 義 を、DSL で grails-app/conf/Config.

groovy に記述することができます。

rabbitmq {

    connectionfactory {

        username = 'guest'

        password = 'guest'

        hostname = 'localhost'

    }

    queues = {

        myQueueName()

        myOtherQueueName autoDelete: false, 

 durable: true, exclusive: false, 

 arguments: [arg1: 'val1', arg2: 'val2']

    }

}

■設定できる項目
両方の定義方法で設定できる項目は、次のとおりです。

項目 説明 デフォルト値

autoDelete クライアントの接
続がなくなった場
合にブローカーか
らキューを削除す
るかどうか

false

durable ブローカの再起動
後もキューが存在
するかどうか

false

exclusive キューを作成した
クライアントのみ
がキューに接続で
きるようにするか
どうか

false

arguments キュー作成時の引
数

null

エクスチェンジの設定
エクスチェンジは、メッセージを受け取る役割を担います。エ

クスチェンジが受け取ったメッセージはバインドに従って、適切
なキューに送られます。

エクスチェンジの設定は、grails-app/conf/Config.groovy で行
います。

rabbitmq {

    connectionFactory {

        ...

    }

    queues = {

        exchange name: 'my.topic', type: topic

    }

}

exchange に設定できる項目は、次のとおりです。

項目 説明 デフォルト値

name エクスチェンジの
名前

なし

type エクスチェンジ
の種類。direct 
fanout, topic, 
headers から指定

なし

autoDelete クライアントの接
続がなくなった場
合にブローカーか
らエクスチェンジ
を削除するかどう
か

false

durable ブローカの再起動
後もエクスチェン
ジが存在するかど
うか

false

キューとエクスチェンジのバインド
エクスチェンジとキューのバインドも、設定で指定できます。

バインドの設定は、grails-app/conf/Config.groovy で行います。

rabbitmq {

    connectionFactory {

        ...

    }

    queues = {

        exchange name: 'my.topic', type: topic, 

 durable: false, {

         foo durable: true, 

  binding: 'shares.#'

          bar durable: false, 

  autoDelete: true, 

     binding: 'shares.nyse.*'

        }

    }

}

上記の例では、'my.topic' という名前のエクスチェンジを 1 つ
を定義し、'foo', 'bar' という名前のキューにバインドします。

どのメッセージをキューにバインドするかは、binding で指定
します。

ルーティングキーが 'shares.' で始まるメッセージは、foo キュー
にバインドされます。

ルーティングキーが 'shares.nyse.' で始まり 1 語続くメッセージ
は、bar キューにバインドされます。

メッセージの送信
メ ッ セ ー ジ の 送 信 は、rabbitSend メ ソ ッ ド で 行 い ま す。

rabbitSend メソッドは、コントローラ、サービス、タグリブといっ
た Grails アーティファクトに RabbitMQ プラグインによって追加
されます。

class MessageController {

    def sendMessage = {

        rabbitSend 'someQueueName', 'someMessage'

        ... 

    }

}

rabbitSend メソッドは、引数を 2 つとります。1 つ目はキュー
の名前で、2 つ目は送信するメッセージそのものです。

メッセージの送信は、org.springframework.amqp.rabbit.core.
RabbitTemplate インスタンスを使う方法もあります。この方法
については、後ほど説明します。

メッセージの受信
メッセージの受信は、次の 2 通りの方法があります：

• エクスチェンジの購読による受信 ( 発行 / 購読モデル )
• キューからの受信

エクスチェンジの購読による受信
エクスチェンジの購読による受信では、まずエクスチェンジの

定義を行います。
Grails アプリケーションが発行者となる場合、grails-app/conf/

Config.groovy でエクスチェンジの定義をします。

次に、メッセージを購読する処理のため、rabbitSubscribe と
いう名前の静的プロパティと、handleMessage() という名前の
メッセージを処理するメソッドを持つサービスを作成します。

package rabbit

class SharesService {

    static rabbitSubscribe = 'shares'

    void handleMessage(message) {

        // handle message

        ...

    }

}

rabbitSubscribe プロパティに指定するのはエクスチェンジの
名前で、指定されたエクスチェンジ宛てに送られたメッセージを
購読することになります。

エクスチェンジが受け取ったメッセージは、handlerMessage()
メ ソ ッ ド の 引 数 と し て 指 定 さ れ、 購 読 対 象 の サ ー ビ ス の
handlerMessage() メソッドが呼び出されます。

topic タイプのエクスチェンジの場合、ルーティングキーを指
定することで、受信するメッセージをコントロールできます。

rabbitSubscribe プロパティをハッシュ形式で指定し、name で
エクスチェンジの名前、routingKey でルーティングキー( のパター
ン ) を指定します。

package rabbit

class SharesService {

    static rabbitSubscribe = [ name: 'shares', 

   routingKey: 'NYSE.GE' ]

    ...

}

キューからの受信
キューからの受信では、まずキューの定義を行います。
次に、メッセージを受信する処理のため、rabbitQueue とい

う名前の静的プロパティと、handleMessage() という名前のメッ
セージを処理するメソッドを持つサービスを作成します。

Grails Plugin 探訪　第 5 回〜 RabbitMQ プラグイン〜

package rabbit

class DemoService {

    static rabbitQueue = 'someQueueName'

    void handleMessage(message) {

        // handle message

        ...

    }

}

rabbitQueue プロパティに指定するのはキューの名前です。
メッセージは handlerMessage() メソッドの引数として指定さ

れ、受信対象のサービスの handlerMessage() メソッドが呼び出
されます。

メッセージ
RabbitMQ プラグインでは、メッセージとして、次の型のオブ

ジェクトを使用することができます：

* 文字列
* バイト配列
* マップ
* シリアライザブル

handleMessage() メソッドをオーバーロードすることで、メッ
セージの型ごとに呼び出すメソッドを変えることができます：

package rabbit

class DemoService {

    static rabbitQueue = 'someQueueName'

    void handleMessage(String textMessage) {

        // handle String message

        ...

    }

    void handleMessage(Map mapMessage) {

        // handle Map message

        ...

    }

    void handleMessage(byte[] byteMessage) {

        // handle byte array message

        ...

    }

}

RabbitTemplate クラスを使用する。
自分で細かい制御をしたい場合、org.springframework.amqp.

rabbit.core.RabbitTemplate ク ラ ス の イ ン ス タ ン ス を Bean と し
て、コントローラやサービスにインジェクションし使用すること
ができます。

Bean の名前は、rabbitTemplate です。

class MessageController {

    def rabbitTemplate

    def sendMessage = {

        rabbitTemplate.convertAndSend('someQueueName', 

            'someMessage)

        ...

    }

}

RabbitTemplate ク ラ ス の 詳 細 は、http://static.springsource.
org/spring-amqp/docs/latest-ga/ を参照してください。

終わりに
アプリケーションで非同期処理を行う際、メッセージシステム

を使うことを、そして RabbitMQ を使うことを検討してみてはい
かがでしょうか。

▼ぐるーびーたん　第 3 話までのあらすじ
プログラマを目指して熊本から上京したぐるーびー

たん。いろいろ覚えて楽しそうなんだけど。。。君の名
は確か !?
※この作品は、たいがいフィクションです。実在の人物、

団体とは関係ありません。

ぐるーびーたん　第 4 話

リリース情報 2011.12.15

Grails
Grails は、Groovy や Hibernate などをベースとしたフルスタック
の Web アプリケーションフレームワークです。
URL: http://grails.org/
バージョン : 1.2.5, 1.3.7, 2.0.0

■更新情報
• 2.0.0 では、ページスコープで <g:include> が <g:set> を壊し

てしまうバグや、"formula" 派生プロパティが不正になるバ
グなど、いくつかバグの改修や、機能の改善が行われていま
す。

• 2.0.0 リリースノート : http://grails.org/2.0.0+Release+Notes
• 山本さんのブログ： 

http://d.hatena.ne.jp/mottsnite/20111215/1323965990

Groovy
Groovy は、JavaVM 上で動作する動的言語です。
URL: http://groovy.codehaus.org/
バージョン : 1.7.10, 1.8.4, 2.0.0-beta-1

■更新情報
• 1.8.4 では、匿名の内部クラスでプロパティにアクセ

スした際に NullPointerException が発生するバグや、
DefaultTypeTransformation.booleanUnbox のパフォーマン
スの改善など、いくつかバグの改修や、機能の改善が行われ
ています。

• 1.8.4 リリースノート : http://jira.codehaus.org/secure/
ReleaseNote.jspa?projectId=10242&version=17852

• 2.0.0-beta-1 では、Grab の簡略表記が正しく動作しないバグ
や、@Immutable が java.util.UUID をサポートする改良など、
いくつかバグの改修や、機能の改善が行われています。

• 2.0.0-beta-1リリースノート : http://jira.codehaus.org/secure/
ReleaseNote.jspa?projectId=10242&version=17925

Griffon
Griffon は、デスクトップアプリケーションを開発するためのア
プリケーションフレームワークです。
URL: http://griffon.codehaus.org/
バージョン : 0.9.4

■更新情報
• 0.9.4 では、ビルド時にソースのエンコーディングを指定で

きるようになったり、griffon.core.ServiceManager 型のイン
スタンスを介して全てのサービスのインスタンスにアクセス
できるようになるなど、いくつかバグの改修や、機能の改善
が行われています。

• 0.9.4 リリースノート :  
http://griffon.codehaus.org/Griffon+0.9.4

Gant
Gant は、XML の代わりに Groovy で Ant タスクを記述し実行する
ビルド管理ツールです。
URL: http://gant.codehaus.org/
バージョン : 1.9.7

■更新情報
• 1.9.7 では、ANT_HOME に空白文字が含まれている場合 gant.

bat が失敗するバグに対応しています。
• 1.9.7 リリースノート : http://jira.codehaus.org/secure/

ReleaseNote.jspa?projectId=11660&version=17496

Gradle
Gradle は、Groovy でビルドスクリプトを記述し実行するビルド
管理ツールです。
URL: http://www.gradle.org/
バージョン : 0.9.2, 1.0-milestone-6

■更新情報
• 1.0-milestone-6 では、依存解決が以前のリリースより速く

なったり、デーモンが改良されたりするなど、いくつかバグ
の改修や、機能の改善が行われています。

• 1.0-milestone-6 リリースノート : http://wiki.gradle.org/
display/GRADLE/Gradle+1.0-milestone-6+Release+Notes

Gaelyk
Gaelyk は、Groovy で記述する Google App Engine for Java 用のラ
イトウェイトなフレームワークです。
URL: http://gaelyk.appspot.com/
バージョン : 1.1

■更新情報
• 1.1では、Groovy が1.8.4、App Engine SDK が1.6.0にアップデー

トされたり、データストアサービスに get() メソッドが追加
されるなど、いくつかバグの改修や、機能の改善が行われて
います。

• 1.1 リリースノート : http://gaelyk.appspot.com/download

Google App Engine SDK for Java
Google App Engine SDK for Java は、Java で Google App Engine
用の Web アプリケーションを開発するための SDK です。
URL: http://code.google.com/intl/ja/appengine/
バージョン : 1.6.1

■更新情報
• 1.6.1 では、プログラムでアプリケーションのログを読み取

ることができるようになったり、ハイレプリケーションデー
タストアの移行ユーティリティが一般利用可能な機能となる
など、いくつかバグの改修や、機能の改善が行われています。

• 1.6.1 リリースノート : http://code.google.com/p/
googleappengine/wiki/SdkForJavaReleaseNotes

GPars
GPars は、Groovy に直感的で安全な並行処理を提供するシステム
です。
URL: http://gpars.codehaus.org/
バージョン : 0.12GA

Groovy++
Groovy++ は、Groovy 言語に対して静的な機能を拡張します。
URL: http://code.google.com/p/groovypptest/
バージョン : 0.9.0

■更新情報
• 0.9.0 では、Groovy 1.8.2 対応されています。
• 0.9.0 リリースメール : http://groups.google.

com/group/groovyplusplus/browse_thread/
thread/237434c427c57f11?pli=1

Spock
Spock は、Java や Groovy 用のテストと仕様のためのフレームワー
クです。
URL: http://code.google.com/p/spock/
バージョン : 0.5

GroovyServ
GroovyServ は、Groovy 処理系をサーバとして動作させることで
groovy コマンドの起動を見た目上高速化するものです。
URL: http://kobo.github.com/groovyserv/
バージョン : 0.9

Geb
Geb は、Groovy を使用した Web ブラウザを自動化する仕組みで
す。
URL: http://www.gebish.org/
バージョン : 0.6.0

Easyb
Easyb は、 ビ ヘ イ ビ ア 駆 動 開 発 (Behavior Driven 
Development:BDD) 用のフレームワークです。
URL: http://www.easyb.org/
バージョン : 0.9.8

Gmock
Gmock は、Groovy 用のモック・フレームワークです。 
URL: http://code.google.com/p/gmock/
バージョン : 0.8.1

HTTPBuilder
HTTPBuilder は、HTTP ベースのリソースに簡単にアクセスするた
めの方法です。
URL: http://groovy.codehaus.org/modules/http-builder/
バージョン : 0.5.1

CodeNarc
CodeNarc は、Groovy 向けの静的コード解析ツールです。
URL: http://codenarc.sourceforge.net/
バージョン : 0.16.1

■更新情報
• 0.16.1 では、Groovy 1.8 で CodeNarc を実行した際に

StackOverflowError が発生するバグに対応しています。
• 0.16.1 リリースノート : http://sourceforge.net/projects/

codenarc/files/codenarc/CodeNarc%20Version%200.16.1/

GMetrics
GMetrics は、Groovy ソースコードのサイズや複雑さを計算した
り報告するためのツールです。
URL: http://gmetrics.sourceforge.net/
バージョン : 0.4

■更新情報
• 0.4 では、Groovy が 1.7 にアップグレードされたり、Groovy 

1.7.7 以上で GMetrics ができないバグなど、いくつかバグの
改修や、機能の改善が行われています。

• 0.4 リリースノート :  
http://sourceforge.net/projects/gmetrics/files/gmetrics-0.4/

GContracts
GContracts は、Groovy で契約プログラミングを行うためのフレー
ムワークです。
URL: https://github.com/andresteingress/gcontracts
バージョン : 1.2.4

GroovyFX
GroovyFX は、JavaFX を Groovy で書きやすくするためのフレー
ムワークです。
URL: http://groovy.codehaus.org/GroovyFX
バージョン : 1.0 Alpha

GBench
GBench は、Groovy のためのベンチマーク・フレームワークです。
URL: http://code.google.com/p/gbench/
バージョン : 0.2.2

Betamax
Betamax は、HTTP 通信の内容を保存し再生するテストツールで
す。
URL: https://github.com/robfletcher/betamax
バージョン : 1.0

リリース情報

http://groups.google.com/group/groovyplusplus/browse_thread/thread/237434c427c57f11?pli=1
http://groups.google.com/group/groovyplusplus/browse_thread/thread/237434c427c57f11?pli=1
http://groups.google.com/group/groovyplusplus/browse_thread/thread/237434c427c57f11?pli=1
http://code.google.com/p/spock/
http://kobo.github.com/groovyserv/
http://www.gebish.org/
http://www.easyb.org/
http://code.google.com/p/gmock/
http://groovy.codehaus.org/modules/http-builder/
http://codenarc.sourceforge.net/
http://sourceforge.net/projects/codenarc/files/codenarc/CodeNarc%20Version%200.16.1/
http://sourceforge.net/projects/codenarc/files/codenarc/CodeNarc%20Version%200.16.1/
http://gmetrics.sourceforge.net/
http://sourceforge.net/projects/gmetrics/files/gmetrics-0.4/
https://github.com/andresteingress/gcontracts
http://groovy.codehaus.org/GroovyFX
http://code.google.com/p/gbench/
https://github.com/robfletcher/betamax


発行人：日本 Grails/Groovy ユーザーグループ
編集長：川原正隆
編集：G* Magazine 編集委員（杉浦孝博、奥清隆）
デザイン：㈱ニューキャスト
表紙：川原正隆
編集協力：JGGUG 運営委員会
Mail：info@jggug.org

G* Magazine vol.4 2011.12
http://www.jggug.org

Publisher：Japan Grails/Groovy User Group
Editor in Chief：Masataka Kawahara
Editors：G* Magazine Editors Team
　　　　　 (Takahiro Sugiura, Kiyotaka Oku)
Design：NEWCAST inc.
CoverDesign：Masataka Kawahara
Cooperation：JGGUG Steering Committee
Mail：info@jggug.org

© 2011 JGGUG  掲載記事の再利用については 
Creative Commons ライセンスによります。

■企業スポンサー ■個人サポーター

須江 信洋 様

関谷 和愛 様

田中　 明 様

個人サポータ制度のお知らせ 

JGGUG では , 2011 年度より個人サポータ制度を始めました .

今まで JGGUG の経済的な基盤はすべて法人会員の年会費に依存していました。 しかし、 個人会員も金銭面からサポート
したいという声があがり、JGGUG 運営基盤の裾野を拡げるためにも、個人サポータという形で個人会員からの寄付を受け
入れることにした次第です。

個人サポータとなっていただいた方には、 一年間にわたって JGGUG が発行する G* Magazine（年数回刊。基本的に電子
版として配布予定）に個人サポータとしてお名前を掲載します（掲載を希望しない旨お申し出いただけば掲載しません）。

個人サポータとなるには、まず supporters@jggug.org にメイルで
• お名前
• 予定金額
G* Magazine へのご芳名掲載の可否
をお知らせください。追って、運営委員より振込先の情報などを返信します。
皆様のサポートをお待ちしております。

日本 Grails/Groovy ユーザーグループ
代表　山田 正樹

mailto:info%40jggug.org?subject=
http://www.jggug.org
mailto:info%40jggug.org?subject=
http://creativecommons.org/licenses/by-sa/2.1/jp/
http://www.delight-tech.co.jp/
http://www.metabolics.co.jp/
http://www.ntts.co.jp/
http://www.xmldo.jp
mailto:supporters%40jggug.org?subject=

	表紙
	目次
	検証! Grails2.0の世界 Part.2	
	GDK48投票アプリの作り方（前編）	
	GroovyFXで遊ぼう（導入編）
	Betamaxことはじめ
	Grails Plugin探訪〜第5回 RabbitMQプラグイン〜
	JGGUG４コマ漫画「ぐるーびーたん」第4話
	リリース情報



